

Lucas Peixoto de Almeida Cavalcante

Especificação da gramática da linguagem de programação LisC

Especificação da gramática da linguagem de programação LisC, definida pelo aluno, para a disciplina de Compiladores, correspondente à parte da avaliação da AB2 do semestre de 2019.1, sob orientação do **Prof. Alcino Dall Igna Jr.**

Universidade Federal de Alagoas

Instituto de Computação

Brasil

Maceió/Al
2019.1

Sumário

1	INTRODUÇÃO	3
2	ALTERAÇÃO NOS TOKENS	4
3	GRAMÁTICA ORIGINAL	5
4	GRAMÁTICA LL(1)	9

1 Introdução

Esse documento descreve a gramática construída da linguagem de programação LisC, que deverá ser respeitada na implementação do analisador sintático em C. O analisador sintático escolhido é o preditivo recursivo. Diante da escolha deste analisador sintático, teremos também nesse documento as alterações necessárias na gramática para que a gramática se torne uma $LL(1)$, pré-requisito obrigatório de todo analisador sintático preditivo recursivo.

2 Alteração nos tokens

A fim de diminuir e simplificar a gramática, foi decidido agrupar alguns tokens, gerando assim novos tokens e eliminando os antigos. O agrupamento pode ser melhor visualizado na **tabela 1**.

Novo token	Relação com tokens eliminados	Expressão regular
OP_ADI	OP_PLUS OP_MINUS	'-' '+'
OP_RELACIONAL	OP_EQUAL OP_DIFF OP_LEQ OP_LESS OP_GEQ OP_GREATER	'==' '!=' '<=' '<' '>=' '>'
OP_LOGIC	OP_OR OP_AND	'&&' ' '
OP_MULT_DIV	OP_MULT OP_DIVIDE	'*' '/'
OP_FORMAT	FORMAT_FIELD FORMAT_DECIMAL	'%' '%%'

Tabela 1 – Agrupamento para a criação de novos tokens.

3 Gramática Original

O código 4.1 apresenta a gramática original da linguagem.

```

1 Programa = ListaDecl
2
3 ListaDecl = ListaDecl Decl
4 ListaDecl = Decl
5
6 Decl = DeclVariaveis
7 Decl = DeclFuncao
8
9 DeclVariaveis = DefTipo ListaVariaveis ';'
10
11 DefTipo = 'defint'
12 DefTipo = 'defbool'
13 DefTipo = 'defchar'
14 DefTipo = 'defstring'
15 DefTipo = 'deffloat'
16
17 ListaVariaveis = ListaVariaveis ',' Variavel
18 ListaVariaveis = Variavel
19
20 Variavel = 'id'
21 Variavel = 'id' ExpConcat
22 Variavel = 'id' '[' ExpConcat ']'
23 Variavel = 'id' '[' ExpConcat ']' '{' ListaUni '}'
24
25 ListaUni = 'const_string'
26 ListaUni = ListaInt
27 ListaUni = ListaChar
28 ListaUni = ListaBool
29 ListaUni = ListaFloat
30
31 ListaInt = ListaInt ',' 'const_int'
32 ListaInt = 'const_int'
33
34 ListaChar = ListaChar ',' 'const_char'
35 ListaChar = 'const_char'
36
37 ListaBool = ListaBool ',' 'const_bool'
38 ListaBool = 'const_bool'
39
40 ListaFloat = ListaFloat ',' 'const_float'
41 ListaFloat = 'const_float'

```

```

42
43 DeclFuncao = 'defun' 'id' '(' ListaParam ')' '(' TipoReturn ')' '{'
      ListaSent '}'
44
45 ListaParam = ListaParam ',' Param
46 ListaParam = Param
47 ListaParam = 'void'
48
49 Param = Tipo 'id'
50 Param = Tipo 'id' '[' ' ' ]'
51
52 TipoReturn = 'void'
53 TipoReturn = Tipo
54
55 ListaSent = ListaSent Sent
56 ListaSent = Sent
57
58 Sent = DeclVariaveis
59 Sent = Comando
60
61 Comando = Continue
62 Comando = Break
63 Comando = Atribuicao
64 Comando = ChamadaFunc
65 Comando = Return
66 Comando = Read
67 Comando = Echo
68 Comando = If
69 Comando = For
70 Comando = While
71
72 Continue = 'continue' ';'
73
74 Break = 'break' ';'
75
76 Atribuicao = 'set' 'id' ExpConcat ';'
77 Atribuicao = 'set' 'id' '[' ExpConcat ']' ExpConcat ';'
78
79 ChamadaFunc = 'id' '(' ListaArg ')' ';'
80
81 Return = 'return' ExpConcat ';'
82
83 Read = 'read' '(' ListaArg ')' ';'
84
85 Echo = 'echo' '(' ExpConcat ')' ';'
86 Echo = 'echo' '(' ExpConcat ')' '(' ListaArg ')' ';'
87

```

```

88 If = 'if' '(' ExpBooleana ')' '{' ListaSent '}' Else
89
90 Else = 'else' '{' ListaSent '}'
91 Else = epsilon
92
93 For = 'for' '(' 'id' ',' ExpAritmetica ',' ExpAritmetica ForPasso ')' '{'
    ListaSent '}'
94
95 While = 'while' '(' ExpBooleana ')' '{' ListaSent '}'
96
97 ForPasso = ',' ExpAritmetica
98 ForPasso = epsilon
99
100 ListaArg = ListaArg ',' 'id'
101 ListaArg = 'id'
102
103 Tipo = 'int'
104 Tipo = 'char'
105 Tipo = 'string'
106 Tipo = 'float'
107 Tipo = 'bool'
108
109 ExpAritmetica = ExpAritmetica 'op_adi' TermAritmetico
110 ExpAritmetica = TermAritmetico
111
112 ExpBooleana = ExpBooleana 'op_logic' TermBooleano
113 ExpBooleana = TermBooleano
114
115 ExpConcat = ExpConcat '<<' ExpAritmetica
116 ExpConcat = ExpBooleana
117
118 TermBooleano = '!' TermBooleano
119 TermBooleano = ExpAritmetica 'op_relacional' ExpAritmetica
120 TermBooleano = ExpAritmetica
121
122 TermAritmetico = TermAritmetico 'op_mult_div' TermFormatacao
123 TermAritmetico = TermFormatacao
124
125 TermFormatacao = TermFormatacao 'op_format' FatorAritmetico
126 TermFormatacao = FatorAritmetico
127
128 FatorAritmetico = '-' FatorAritmetico
129 FatorAritmetico = 'const_float'
130 FatorAritmetico = 'const_char'
131 FatorAritmetico = 'const_bool'
132 FatorAritmetico = 'const_int'
133 FatorAritmetico = 'const_string'

```

```
134 FatorAritmetico = '(' ExpBooleana ')'  
135 FatorAritmetico = 'id'  
136 FatorAritmetico = 'id' '(' ListaArg ')'  
137 FatorAritmetico = 'id' '[' ExpAritmetica ']'
```

Código 3.1 – Gramática original da linguagem de programação LisC.

4 Gramática LL(1)

```
1 Programa = ListaDecl
2
3 ListaDecl = Decl ListaDeclR
4
5 ListaDeclR = Decl ListaDeclR
6 ListaDeclR = epsilon
7
8 Decl = DeclVariaveis
9 Decl = DeclFuncao
10
11 DeclVariaveis = DefTipo ListaVariaveis ';'
12
13 DefTipo = 'defint'
14 DefTipo = 'defbool'
15 DefTipo = 'defchar'
16 DefTipo = 'defstring'
17 DefTipo = 'deffloat'
18
19 ListaVariaveis = Variavel ListaVariaveisR
20
21 ListaVariaveisR = ',' Variavel ListaVariaveisR
22 ListaVariaveisR = epsilon
23
24 Variavel = 'id' VariavelF
25
26 VariavelF = ExpConcat
27 VariavelF = epsilon
28 VariavelF = '[' ExpConcat ']' VariavelFF
29
30 VariavelFF = '{' ListaUni '}'
31 VariavelFF = epsilon
32
33 ListaUni = 'const_string'
34 ListaUni = ListaInt
35 ListaUni = ListaChar
36 ListaUni = ListaBool
37 ListaUni = ListaFloat
38
39 ListaInt = 'const_int' ListaIntR
40
41 ListaIntR = ',' 'const_int' ListaIntR
42 ListaIntR = epsilon
43
```

```
44 ListaChar = 'const_char' ListaCharR
45
46 ListaCharR = ',' 'const_char' ListaCharR
47 ListaCharR = epsilon
48
49 ListaBool = 'const_bool' ListaBoolR
50
51 ListaBoolR = ',' 'const_bool' ListaBoolR
52 ListaBoolR = epsilon
53
54 ListaFloat = 'const_float' ListaFloatR
55
56 ListaFloatR = ',' 'const_float' ListaFloatR
57 ListaFloatR = epsilon
58
59 DeclFuncao = 'defun' 'id' '(' ListaParam ')' '(' TipoReturn ')' '{'
    ListaSent '}'
60
61 ListaParam = Param ListaParamR
62 ListaParam = 'void'
63
64 ListaParamR = ',' Param ListaParamR
65 ListaParamR = epsilon
66
67 Param = Tipo 'id' ParamF
68
69 ParamF = '[' ']'
70 ParamF = epsilon
71
72 TipoReturn = 'void'
73 TipoReturn = Tipo
74
75 ListaSent = Sent ListaSentR
76
77 ListaSentR = Sent ListaSentR
78 ListaSentR = epsilon
79
80 Sent = DeclVariaveis
81 Sent = Comando
82
83 Comando = Continue
84 Comando = Break
85 Comando = Atribuicao
86 Comando = ChamadaFunc
87 Comando = Return
88 Comando = Read
89 Comando = Echo
```

```

90 Comando = If
91 Comando = For
92 Comando = While
93
94 Continue = 'continue' ';'
95
96 Break = 'break' ';'
97
98 Atribuicao = 'set' 'id' AtribuicaoF
99
100 AtribuicaoF = ExpConcat ';'
101 AtribuicaoF = '[' ExpConcat ']' ExpConcat ';'
102
103 ChamadaFunc = 'id' '(' ListaArg ')' ';'
104
105 Return = 'return' ExpConcat ';'
106
107 Read = 'read' '(' ListaArg ')' ';'
108
109 Echo = 'echo' '(' ExpConcat ')' EchoF
110
111 EchoF = '(' ListaArg ')' ';'
112 EchoF = ';'
113
114 If = 'if' '(' ExpBooleana ')' '{' ListaSent '}' Else
115
116 Else = 'else' '{' ListaSent '}'
117 Else = epsilon
118
119 For = 'for' '(' 'id' ',' ExpAritmetica ',' ExpAritmetica ForPasso ')' '{'
    ListaSent '}'
120
121 While = 'while' '(' ExpBooleana ')' '{' ListaSent '}'
122
123 ForPasso = ',' ExpAritmetica
124 ForPasso = epsilon
125
126 ListaArg = 'id' ListaArgR
127
128 ListaArgR = ',' 'id' ListaArgR
129 ListaArgR = epsilon
130
131 Tipo = 'int'
132 Tipo = 'char'
133 Tipo = 'string'
134 Tipo = 'float'
135 Tipo = 'bool'

```

```

136
137 ExpAritmetica = TermAritmetico ExpAritmeticaR
138
139 ExpAritmeticaR = 'op_adi' TermAritmetico ExpAritmeticaR
140 ExpAritmeticaR = epsilon
141
142 ExpBooleana = TermBooleano ExpBooleanaR
143
144 ExpBooleanaR = 'op_logic' TermBooleano ExpBooleanaR
145 ExpBooleanaR = epsilon
146
147 ExpConcat = ExpBooleana ExpConcatR
148
149 ExpConcatR = '<<' ExpAritmetica ExpConcatR
150 ExpConcatR = epsilon
151
152 TermBooleano = '!' TermBooleano
153 TermBooleano = ExpAritmetica TermBooleanoF
154
155 TermBooleanoF = 'op_relacional' ExpAritmetica
156 TermBooleanoF = epsilon
157
158 TermAritmetico = TermFormatacao TermAritmeticoR
159
160 TermAritmeticoR = 'op_mult_div' TermFormatacao TermFormatacaoR
161 TermAritmeticoR = epsilon
162
163 TermFormatacao = FatorAritmetico TermFormatacaoR
164
165 TermFormatacaoR = 'op_format' FatorAritmetico TermFormatacaoR
166 TermFormatacaoR = epsilon
167
168 FatorAritmetico = '-' FatorAritmetico
169 FatorAritmetico = 'const_float'
170 FatorAritmetico = 'const_char'
171 FatorAritmetico = 'const_bool'
172 FatorAritmetico = 'const_int'
173 FatorAritmetico = 'const_string'
174 FatorAritmetico = '(' ExpBooleana ') '
175 FatorAritmetico = 'id' FatorAritmeticoF
176
177 FatorAritmeticoF = '(' ListaArg ') '
178 FatorAritmeticoF = '[' ExpAritmetica ']'
179 FatorAritmeticoF = epsilon

```

Código 4.1 – Gramática original da linguagem de programação LisC.